



# Integrated Circuits Design by FPGA

م.م. أحمد مؤيد عبدالحسين  
جامعة الفرات الأوسط التقنية / الكلية التقنية الهندسية / نجف

# Lecture 5

## VHDL Sequential Code

# Objectives of this Lecture

- To understand what is the Sequential VHDL code.
- Also, this lecture is very important, for it allows a better understanding of where the parallel VHDL code or sequential VHDL code, as well as the consequences of using one or the other.

# Contents of this Lecture

- Introduction about Sequential VHDL code.
- Sequential VHDL code inside **Processes**
- Sequential VHDL code inside **Processes**\ **IF**
- Sequential VHDL code inside **Processes**\ **WAIT**
- Sequential VHDL code inside **Processes**\ **CASE**

# Introduction about Sequential VHDL code

- VHDL code can be concurrent (parallel) or sequential.
- As mentioned in lecture 4, VHDL code is inherently concurrent.
- **PROCESSES, FUNCTIONS, and PROCEDURES** are the only sections of code that are executed sequentially. However, as a whole, any of these blocks is still concurrent with any other statements placed outside it .

# Introduction about Sequential VHDL code

```
dev_to_test: shift_reg  
    port map(A, B, C, D, data_in, reset, clk);
```

```
clk_stimulus: process  
begin  
    wait for 10 ns;  
    clk <= not clk;  
end process clk_stimulus;
```

```
data_stimulus: process  
begin  
    wait for 40 ns;  
    data_in <= not data_in;  
    wait for 150 ns;  
end process data_stimulus;
```

**Concurrent  
statements**

```
end test;
```

# Introduction about Sequential VHDL code

- The statements discussed in this section are all sequential, that is, allowed only inside PROCESSES, FUNCTIONS, or PROCEDURES. They are: IF, WAIT, CASE, and LOOP.
- VARIABLES are also restricted to be used in sequential code only (that is, inside a PROCESS, FUNCTION, or PROCEDURE).

# Sequential VHDL code inside Processes

- A PROCESS is a sequential section of VHDL code. It is characterized by the presence of **IF**, **WAIT**, **CASE**, or **LOOP**, and by a sensitivity list (except when WAIT is used).
- A PROCESS must be installed in the main code, and is executed every time a signal in the **sensitivity list changes** (or the **condition related to WAIT is fulfilled**). Its syntax is shown below.

```
[label:] PROCESS (sensitivity list)
  [VARIABLE name type [range] [:= initial_value;]]
BEGIN
  (sequential code)
END PROCESS [label];
```




# Sequential VHDL code inside Processes \IF

## IF structure

```
IF conditions THEN assignments;  
ELSIF conditions THEN assignments;  
...  
ELSE assignments;  
END IF;
```

# Sequential VHDL code inside Processes \IF

## Example 6.1




**Figure 6.1**  
DFF with asynchronous reset of example 6.1.

# Sequential VHDL code inside Processes \IF

```
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY dff IS
6      PORT (d, clk, rst: IN STD_LOGIC;
7            q: OUT STD_LOGIC);
8  END dff;
9  -----
10 ARCHITECTURE behavior OF dff IS
11 BEGIN
12     PROCESS (clk, rst)
13     BEGIN
14         IF (rst='1') THEN
15             q <= '0';
16         ELSIF (clk'EVENT AND clk='1') THEN
17             q <= d;
18         END IF;
19     END PROCESS;
20 END behavior;
```

# Sequential VHDL code inside Processes \IF




**Figure 6.2**  
Simulation results of example 6.1.

**DFF : D Flip Flop**

# Sequential VHDL code inside Processes \IF

## Example 6.2




**Figure 6.3**  
Counter of example 6.2.

# Sequential VHDL code inside Processes \IF

```
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY counter IS
6      PORT (clk : IN STD_LOGIC;
7            digit : OUT INTEGER RANGE 0 TO 9);
8  END counter;
9  -----
10 ARCHITECTURE counter OF counter IS
11 BEGIN
12     count: PROCESS(clk)
13         VARIABLE temp : INTEGER RANGE 0 TO 10;
14     BEGIN
15         IF (clk'EVENT AND clk='1') THEN
16             temp := temp + 1;
17             IF (temp=10) THEN temp := 0;
18             END IF;
19         END IF;
20         digit <= temp;
21     END PROCESS count;
22 END counter;
```

# Sequential VHDL code inside Processes \IF



**Figure 6.4**  
Simulation results of example 6.2.

**Counter**

# Sequential VHDL code inside Processes \IF

Example 6.3: shift register

```
1  -- Library's
2  library IEEE;
3  use IEEE.STD_LOGIC_1164.ALL;
4  use IEEE.numeric_std.all;
5
6  -- Entity Declaration
7  entity Shift_Reg is
8  port (
9      A      : out std_logic; -- single bit in register
10     B      : out std_logic; -- single bit in register
11     C      : out std_logic; -- single bit in register
12     D      : out std_logic; -- single bit in register
13     data_in : in  std_logic; -- data input (1 or 0)
14     reset   : in  std_logic; -- when this signal goes high clear
15     -- all bit values on a clock cycle
16     clk     : in  std_logic); -- input clock
17 end Shift_Reg;
18
19 -- Architecture Body
20 architecture behavior of Shift_Reg is
21
22 -- Defined Signals used in Architecture
23 signal A_reg, B_reg : std_logic := '0';
24 signal C_reg, D_reg : std_logic := '0';
25
```

```
26 -- Begin Architecture
27 begin
28
29 -- Signal Assignments
30 A <= A_reg;
31 B <= B_reg;
32 C <= C_reg;
33 D <= D_reg;
34
35 -- Process that is used to shift values
36 -- ** HINT **
37 -- (We want this process to be evaluated
38 -- on every clock cycle)
39 reg_process: process(clk)
40 begin
41     if(rising_edge(clk)) then
42         if(reset = '1') then
43             A_reg <= '0';
44             B_reg <= '0';
45             C_reg <= '0';
46             D_reg <= '0';
47         else
48             -- ** HINT **
49             -- This is where the shifting actually occurs
50             -- depending on how you code this, you can have
51             -- a shift right or shift left register
52             A_reg <= data_in;
53             B_reg <= A_reg;
54             C_reg <= B_reg;
55             D_reg <= C_reg;
56         end if;
57     end if;
58 end process reg_process;
59 end behavior;
```



# Sequential VHDL code inside Processes \IF



Example 6.3: shift register

# Sequential VHDL code inside Processes \WAIT

(there are three forms of WAIT) is shown below.

```
WAIT UNTIL signal_condition;
```

```
WAIT ON signal1 [, signal2, ... ];
```

```
WAIT FOR time;
```

# Sequential VHDL code inside Processes \WAIT

Example: 8-bit register with **synchronous** reset.

```
PROCESS          -- no sensitivity list
BEGIN
    WAIT UNTIL (clk'EVENT AND clk='1');
    IF (rst='1') THEN
        output <= "00000000";
    ELSIF (clk'EVENT AND clk='1') THEN
        output <= input;
    END IF;
END PROCESS;
```

# Sequential VHDL code inside Processes \WAIT

Example: 8-bit register with asynchronous reset.

```
PROCESS
BEGIN
    WAIT ON clk, rst;
    IF (rst='1') THEN

        output <= "00000000";
    ELSIF (clk'EVENT AND clk='1') THEN
        output <= input;
    END IF;
END PROCESS;
```

# Sequential VHDL code inside Processes \WAIT

- Finally, **WAIT FOR** is intended for simulation only (waveform generation for testbenches). Example:

**WAIT FOR 5ns;**

# Sequential VHDL code inside Processes \CASE

- **CASE** is another statement intended exclusively for sequential code (along with IF, LOOP, and WAIT). Its syntax is shown below:

```
CASE identifier IS  
    WHEN value => assignments;  
    WHEN value => assignments;  
    ...  
END CASE;
```

- The **CASE** statement (sequential) is very similar to **WHEN** (combinational).

Note : Parallel = Concurrent = Combinational



# Sequential VHDL code inside Processes \CASE

- Here too all permutations must be tested.
- However, **CASE** allows multiple assignments for each test condition (as shown in the previous slide), while **WHEN** allows only one.
- Like in the case of **WHEN** (lecture 4), here too “**WHEN value**” can take up three forms:

```
WHEN value                -- single value
WHEN value1 to value2    -- range, for enumerated data types
                           -- only
WHEN value1 | value2 |... -- value1 or value2 or ...
```

# Sequential VHDL code inside Processes \CASE

**Example 6.7:**  
2-digit counter



Input: "xabcdefg"

**Figure 6.7**  
2-digit counter of example 6.7.

a	b	c	d	e	f	g
1	1	1	1	1	1	1




# Sequential VHDL code inside Processes \CASE

```
2  LIBRARY ieee;
3  USE ieee.std_logic_1164.all;
4  -----
5  ENTITY counter IS
6      PORT (clk, reset : IN STD_LOGIC;
7            digit1, digit2 : OUT STD_LOGIC_VECTOR (6 DOWNTO 0));
8  END counter;
9  -----
10 ARCHITECTURE counter OF counter IS
11 BEGIN
12     PROCESS(clk, reset)
13         VARIABLE temp1: INTEGER RANGE 0 TO 10;
14         VARIABLE temp2: INTEGER RANGE 0 TO 10;
15     BEGIN
16         ---- counter: -----
17         IF (reset='1') THEN
18             temp1 := 0;
19             temp2 := 0;
20         ELSIF (clk'EVENT AND clk='1') THEN
21             temp1 := temp1 + 1;
22             IF (temp1=10) THEN
23                 temp1 := 0;
24                 temp2 := temp2 + 1;
25                 IF (temp2=10) THEN
26                     temp2 := 0;
```



```
27         END IF;
28     END IF;
29 END IF;
30 ---- BCD to SSD conversion: -----
31 CASE temp1 IS
32     WHEN 0 => digit1 <= "1111110";    --7E
33     WHEN 1 => digit1 <= "0110000";    --30
34     WHEN 2 => digit1 <= "1101101";    --6D
35     WHEN 3 => digit1 <= "1111001";    --79
36     WHEN 4 => digit1 <= "0110011";    --33
37     WHEN 5 => digit1 <= "1011011";    --5B
38     WHEN 6 => digit1 <= "1011111";    --5F
39     WHEN 7 => digit1 <= "1110000";    --70
40     WHEN 8 => digit1 <= "1111111";    --7F
41     WHEN 9 => digit1 <= "1111011";    --7B
42     WHEN OTHERS => NULL;
43 END CASE;
44 CASE temp2 IS
45     WHEN 0 => digit2 <= "1111110";    --7E
46     WHEN 1 => digit2 <= "0110000";    --30
47     WHEN 2 => digit2 <= "1101101";    --6D
48     WHEN 3 => digit2 <= "1111001";    --79
49     WHEN 4 => digit2 <= "0110011";    --33
50     WHEN 5 => digit2 <= "1011011";    --5B
51     WHEN 6 => digit2 <= "1011111";    --5F
52     WHEN 7 => digit2 <= "1110000";    --70
53     WHEN 8 => digit2 <= "1111111";    --7F
54     WHEN 9 => digit2 <= "1111011";    --7B
55     WHEN OTHERS => NULL;
56 END CASE;
57 END PROCESS;
58 END counter;
```

# Sequential VHDL code inside Processes \CASE



**Figure 6.8**  
Simulation results of example 6.7.

**2-digit counter**

# Assignments

- Design the same **DFF** of example 6.1 by using **WAIT** statement.
- Design the same progressive 1-digit decimal **counter** of example 6.2 by using **WAIT** statement.

**End of lecture 5**

**Any Questions ?**